Chapter 5. Varying Variables

How many people are in your family? In your class? Are they all the same? Or are they Variable?



Got one in on you, didn't I? There's that word, "Variable."

How many of your friends have the same color hair? The same color eyes? How many are the same age? How many were born in the same month as you? Anyone born on the same day?

How many things about you and your group are the same? How many are "variable?"



Variables in Logo

Now let’s talk about Logo.

By now, you should know what this procedure will look like after it's been run. What do you think?

```
TO BOXES
REPEAT 4 [FD 100 RT 90]
RT 90 PU FD 120 PD LT 90
REPEAT 4 [FD 100 RT 90]
END
```

Sure, that's a procedure to draw two boxes side by side.

But what if you wanted to draw 20 boxes? What if you want each box to be bigger than the last?



What if you want them smaller? In other words, what if you want to vary the size or the number of boxes?

No problem! This is where those things called “variables” come in. A variable is something you put into a procedure so you can change the procedure every time you run it.

Yes, that does sound confusing, doesn't it?

To help explain it, let's take another look at the experiment you did creating pictures using just one shape. Find a big sheet of paper and draw a picture using your favorite shape. Use triangles, squares, or rectangles — or even circles, if you've peeked ahead in this book.

Remember, you can only use one type of shape. But you can vary the size of the shape all you want.

There's that word again, "vary."

Remember the caterpillar example? That's a picture drawn using just squares (and a little piece of a straight line). And don't forget the cat.

OK, got your drawing done? Before you try to put your picture on the computer, let's take a look at the new BOXES procedure below. It may give you some ideas.

```
TO BOXES :SIZE
REPEAT 4 [FD :SIZE RT 90]
RT 90 PU FD :SIZE + 20 PD LT 90
REPEAT 4 [FD :SIZE RT 90]
END
```

You probably know what the variable is, don't you? It's the :SIZE. That's right.

Now when you type BOXES to run the procedure, you have to provide something new, an input.



Try it out. Type...

BOXES 20

BOXES 40

BOXES 60

BOXES 100

When you type BOXES 20, you tell the :SIZE variable to use the :SIZE of 20. What about BOXES 60. What will :SIZE be then?

Variables must always have an input, or value. And they must also have the two dots in front so Logo knows it’s a variable.

Yes, that's a colon. But in Logo, we call them “dots.” You'll find they can save you a lot of time and typing.

Take a look. Remember the TRI procedure? Let's add a variable.

```
TO TRI :N
REPEAT 3 [FD :N RT 120]
END
```

See! You can name variables just about anything you want. Rather than call this one :SIZE, call it :N. The :N can stand for number. Of course, you could call it :X, :Z, or :WHATEVER.



But you still use the dots. You have to do that.

Here are some examples that a 7-year-old had fun dreaming up. They use this SQUARE procedure.

```
TO SQUARE :N
REPEAT 4 [FD :N RT 90]
END
```

It started as a simple exercise to see what different squares would look like.

```
TO SQUARES
SQUARE 60
SQUARE 80
SQUARE 100
SQUARE 120
END
```



Then she added a left turn, and that reminded her of her mom's stacking tables.

```
TO TABLES
SQUARES
LT 90
SQUARES
END
```



The more she looked at the tables, the more it looked like half of a decorative mirror.

```
TO MIRROR
TABLES
LT 90
TABLES
END
```



And what would happen if you stacked mirrors?

```
TO MIRRORS
MIRROR
LT 45
MIRROR
END
```



This is a lot to think about. So why not stop for a while and experiment using one shape in a design.

After you've had fun with one shape, try doing something with two shapes.

You've already seen what you can do with a square and a triangle. These were combined to make a house. Then they were used to make a wheel.

Since you've also made some flowers, maybe you can "plant another garden?"



Polygons and Things

Polygon? Now there's a new word for you. Do you know what it means? No, it doesn't mean that Poly flew away.



We'll talk lots more about polygons. But for now, think about this for a moment.

Squares, triangles, and rectangles are polygons. So are pentagons, hexagons, and octagons.

All of these shapes have one thing in common. They all enclose an area that has at least three sides. (You can't enclose anything with two sides, can you?)

Triangles have three sides, squares and rectangles have four, pentagons have five, and octagons have eight.

A polygon is a closed shape with at least three sides.

Remember the review you did at the end of Chapter 2? You added up the angles used to make squares, triangles, and rectangles. What was the answer?

They all added up to 360, right?

Remember Morf's Rabbit Trail about the clock? How many degrees are in the clock face? There are 360, right? Well, remember that number as we talk about polygons.

Rabbit Trail 16. Variable String Toss

Here's something else to explore. How about trying a variation of the game String Toss? It's called FD :N. (We're sneaking the variables in here, too.) The idea is to create a design by passing the ball of string back and forth. The :N variable can equal one step or as many as you want.

Let's say you want to create a square of string. That's really easy. One person plays the Turtle starting at Home. The Turtle holds one end of the string, gives the ball of string to the first person, and says FD :N times 5. The first person takes 5 steps.

The first person then turns RT 90, holds the string to make a corner, and gives the ball of string to the second person. That person goes FD :N * 5 and RT 90. A third person takes the ball of string and goes FD :N * 5 RT 90. And finally a fourth person takes the string and brings it HOME.

See how this works? The string is now in the shape of a variable square. Now try a hexagon, why don't you?

Then maybe you can connect six triangles to make a fancy hexagon.

It's more fun when you make crazy shapes. Try it.

If you find it hard to see the shapes, have everyone carefully put the string on the floor and then step back. Can you see the shape now?

Hexagons and Spiderwebs

To make that String Toss Game design on the computer, you can use the TRI :N procedure you wrote earlier in this chapter.

```
TO TRI :N
REPEAT 3 [FD :N RT 120]
END
```

What would happen if you repeated the TRI :N procedure, turning after each triangle?

```
REPEAT 6 [TRI :N RT 60]
```



What do you call a shape that has six sides like this? That's a hexagon, right?

Hmmmm? That sort of looks like a see-through box — one of those optical illusions.

But back to hexagons for now.

```
TO HEXAGON :N
REPEAT 6 [TRI :N RT 60]
END
```

Be sure to tell the turtle how big to make the hexagon. Try this:



```
HEXAGON 60
HEXAGON 80
HEXAGON 100
```

What does this look like? Of course, it's a spiderweb!



Can you think of another way to write this procedure so that the turtle will do the same thing? How about this!

```
TO SPIDERWEB :N
HEXAGON :N
HEXAGON :N + 20
HEXAGON :N + 40
END
```

Go ahead. Type the SPIDERWEB :N procedure and then try

```
SPIDERWEB 40
```

Play around with this idea to see what it can do. Make up some other shape procedures using variables.

Adding More Variables

Can you think of a way to use more variables in the SPIDERWEB procedure? What about substituting a variable for 10? For 20? For both?

```
TO SPIDERWEB :N :X :Y
HEXAGON :N
HEXAGON :N + :X
HEXAGON :N + :X * :Y
END
```

This is getting complicated.

:N gives you the size of each side.
:X tells you how much to add to :N
:Y tells you to multiply :X by this number

After you've typed in this procedure, see what happens when you try

```
SPIDERWEB 60 20 2
```

Does this look like the first spiderweb the turtle drew? It should. Let's change the variables to numbers and take a look.

```
TO SPIDERWEB 60 20 2
HEXAGON 60
HEXAGON 60 + 20
HEXAGON 60 + 20 * 2
END
```

Changing a Variable

Typing SPIDERWEB 60 20 2 is fine when you want to make three hexagons that have sides of 60, 80, and 100. But what if you want to do five hexagons? Seven hexagons? Seventy hexagons?

Let's try something! When you write a procedure, it becomes another command you can use, right?

OK. Then let's make the most of it. Tell SPIDERWEB to draw a hexagon using the variable :N. Then tell SPIDERWEB to add 10 to itself and do the same thing again.

```
TO SPIDERWEB :N
HEXAGON :N
SPIDERWEB :N + 10
END
```

Try it! What happens?

Wait a minute!

The last line of the SPIDERWEB procedure has the procedure using itself. That's strange!

No, that isn't strange, that's recursion. There's a whole chapter on what you can do with recursion. For now, let's stick with variables.

Local and Global Variables

Most versions of Logo use two types of variables: local and global. Global variables are used by any procedure. Take a look.

```
TO SHAPES :N
TRI :N
SQUARE :N
RECTANGLE :N
END
```

How about it? Can you write procedures for a triangle, a square, and a rectangle using :N to represent the distance forward.

```
TO TRI :N
REPEAT 2 [FD :N RT 120]
END

TO SQUARE :N
REPEAT 4 [FD :N RT 90]
END

TO RECTANGLE :N
REPEAT 2 [FD :N RT 90 FD :N * 2 RT 90]
END
```

If you type SHAPES 100, each of the procedures will use 100 wherever there is an :N. The :N is a global variable. It's available to anyone who wants to use it.

Global variables tend to be a nuisance. Logo has to keep track of which procedures uses which global variable, what the value of the variable is, has it changed? This takes up valuable memory.

Of course, sometimes you have to use global variables. But it's better if you can use local variables.

Local variables are "local" to the one procedure where it is used. So there isn't nearly as much record-keeping required, making it easier on Logo.

You write them like this:

```
TO TRI
LOCAL "X
MAKE "X 100
REPEAT 3 [FD :X RT 120]
END
```

Go ahead. Change your TRI procedure and then run the SHAPES procedure using SHAPES 100 again. Now what does the picture look like? Why?

You'll see lots more examples of local variables as you move through the rest of this book.

Outputting Variables

OK, local variables are good. Global variables are not so good. Is there another way to pass information between procedures without using global variables?

Sure is!

You can OUTPUT them. You remember, OUTPUT sends information to another procedure. Let's use the TRI procedure as an example. Here's what you need to do.

```
TO TRI
REPEAT 3 [FD X RT 120]
END

TO X
OUTPUT 100
END
```

In this example, X isn't really a variable. So how would you add a local variable to this so that X would pass information to TRI?

How about this?

```
TO X
LOCAL "Z
MAKE "Z READWORD
OUTPUT :Z
END
```

Is this really the best way to run the TRI procedure? Of course not. The important lesson here is

Don't ever close your mind to new possibilities!

Making Variables

LOCAL "X is easy enough to figure out in the TRI procedure. But what's with the MAKE "X 100?

MAKE is a command that gives a value to the variable named "name." The name of a variable must always be what Logo sees as a word. That means it can be a letter, such as :X, or a word, such as :VARIABLE. Here's how it works.

MAKE "<*name*> <*value*>

In the example on the last page, the goal was to MAKE the variable X have the value of 100. Then you can use the variable :X within that procedure whenever you want something to be equal to (have the value of) 100. In the TRI procedure, the variable :N was used as the side of the triangle, which in this case is 100 turtle steps long.

More Ways to Make Variables

You just got introduced to MAKE. Well, Logo gives you lots of other ways to vary your variables. Let's start with another look at MAKE.

```
MAKE "JOE 2
MAKE "TOM 4
MAKE "SAM :JOE + :TOM
```

So what does :SAM equal? If you said six, you get a Gold Star.

You can also NAME :JOE + :TOM "SAM

This does the same thing as MAKE "SAM :JOE + :TOM except that you NAME <*value*> "<*name*>.

If you want to see what :SAM equals, you can tell the computer to

```
PRINT :SAM
or
SHOW :SAM
```

You can also tell Logo to

```
SHOW THING "SAM
or
PRINT THING "SAM
```

THING does the same thing as the dots. It outputs the value of the variable named in the word that follows THING. Sure, that sounds confusing. Try it a few times and it will begin to make sense. That's why Morf likes to experiment so much.

Conditional Things

Remember the SPIDERWEB procedure?

```
TO SPIDERWEB :N
HEXAGON :N
SPIDERWEB :N + 10
END
```

The problem with this procedure is that it just keeps running, filling your screen with spiderwebs. Is there no way to stop it other than pressing the HALT button?

Well, there is a way. You just tell the turtle that IF the last hexagon that it drew was as big as you want the spiderweb to be, THEN stop drawing.

Here's how you use IF. Since IF knows what you mean, you don't have to use the word THEN.

```
TO SPIDERWEB :N
IF :N > 100  [STOP]
HEXAGON :N
SPIDERWEB :N + 10
END
```

Look at that first line in this new procedure. When the turtle reads this line, it learns that IF :N is greater than 100, then stop drawing.

Greater Than, Less Than

That thing that looks like an arrowhead after the :N > is the symbol for "greater than." It means that if the value of :N is greater than 100, then STOP.

If > means "greater than," what does that other arrow symbol [<] mean?

You guessed it. It means "less than." An easy way to remember which symbol is which is that the arrow always points to the smaller value.

- IF :N > 100 means that the value of :N must be larger than 100, at least 101.
- IF :N < 100 means that the value of :N must be less than 100, no more than 99.

For our example, we picked 100 as a place to stop. You can select your own stopping point. Or you can make the stopping point another variable. How would you do that?

Go ahead. Give it a try. But remember, if you're going to use a variable like this, you have to add it to the procedure name.

```
TO SPIDERWEB :N ____
IF :N > ____ [STOP]
HEXAGON :N
SPIDERWEB :N + 10  ____
END
```

"OK, I understand IF. IF something is true, then Logo will carry out the next instruction. And that sits inside brackets. But what if that something is not true? What if I want Ernestine to do something if the answer is false?

_____

TEST

Actually, there are two ways to handle that. Look at how SPIDERWEB has been changed below.

```
TO SPIDERWEB :N
TEST :N > 100
IFTRUE [CS CT PR [SORRY!] STOP]
IFFALSE [HEX :N]
SPIDERWEB :N + 10  ____
END
```

The first lines says to test :N to see if it is greater than 100. The next line says that if the test is true, clear the screen, clear the text, print SORRY!, and stop. The third line says that if :N is not greater than 100, go ahead and run HEX :N. (HEX :N is a new short name for HEXAGON :N.)

What do you think would happen if you left out IFFALSE? Then you'd have

```
TEST :N > 100
IFTRUE [CS CT PR [SORRY!] STOP]
HEX :N
```

Would that work? Try it and see. What did you learn from that?

You don't always have to have both IFTRUE (IFT for short) and IFFALSE (IFF for short) in your procedures.

IFELSE

Another way is to use the IFELSE command. Let's change the SPIDERWEB procedures and try it out.

```
TO SPIDERWEB :N
IFELSE :N > 100 [CS CT PR [SORRY!]STOP][HEX :N]
```

```
SPIDERWEB :N
END
```

The first line says that if :N is greater than 100

- clear the screen
- clear the text
- print SORRY!
- Stop

If not, run the HEX procedure and move on the next line. You can think of IFELSE as

IF a condition is true, THEN do this or ELSE do this. Go ahead and explore. You'll see more of IFELSE.

When you've finished with spiderwebs, why not add variables to your procedures for drawing other shapes? See what you can do with squares, rectangles and things.

Remember, this book is about Discovery!

More on Tessellations

Tessellations are really great places to use variables. These repeating patterns usually start with a basic shape that is repeated in varying sizes.

Do you remember the tessellation that used Diamonds? This gets a bit tricky so think this one through carefully. Can you combine DIAMOND, DIAMOND1, and DIAMOND2 to make one procedure using variables? How would this change the other procedures?

Here are the Diamond procedures.

```
TO DIAMOND
REPEAT 2 [FD 8 RT 60 FD 8 RT 120]
END

TO DIAMOND1
REPEAT 2 [FD 24 RT 60 FD 24 RT 120]
END

TO DIAMOND2
REPEAT 2 [FD 40 RT 60 FD 40 RT 120]
END
```

Look at the distances the turtle moves. Can you write one procedure for these that uses a distance variable?

```
TO DIAMOND :DIST
REPEAT 2 [FD :DIST RT 60 FD :DIST RT 120]
END
```

Now, rather than use DIAMOND, DIAMOND1, or DIAMOND2, you can use DIAMOND 8, DIAMOND 24, or DIAMOND 40.

More Fun With Squares

Let's try a tessellation with squares. The first thing to do is draw a tower of squares, each square smaller than the last.

```
TO SQUARES :S
IF :S < 0 THEN STOP
REPEAT 4 [FD :S RT 90]
FD :S
SQUARES :S - 5
END
```



Try SQUARES now using different inputs. This is going to be the basic pattern in the tessellation. The picture above was made using 50 as the input to SQUARES.

Next, let's make a TOWER of SQUARES.

TOWER takes two inputs: one that says how big the SQUARES are, and the second to tell the turtle how many times to repeat the SQUARES pattern.

```
TO TOWER :S :T
IF :T = 0 THEN STOP
SQUARES :S
TOWER :S :T - 1
END
```



Here's the pattern made by using TOWER 15 5.

And this raises a question. Are you just going to make a tall, skinny tessellation? Or can you make the TOWER procedure turn the corner, maybe like a picture frame?

```
TO FRAME
PU LT 90 FD 100 RT 90 BK 40 PD
REPEAT 4 [TOWER 15 4 RT 90]
END
```

The first thing the FRAME procedure does is move the turtle over to the left. Then it draws the FRAME using the REPEAT command.



```
REPEAT 4 [TOWER 15 4 RT 90]
```

Now we're getting some where. Try different inputs. TOWER 15 4 seems to work pretty good.

To make this into an interesting tessellation, why not just fill up a frame with the SQUARES pattern? How are you going to do that?

_____

Rabbit Trail 17. Tessellating Squares

Here's a quick and easy Rabbit Trail for you. It's a great way to discover what you can do with your SQUARES pattern. You can either use squares of different sizes or better yet, print a page full of the SQUARES pattern and cut them out.

Now move the patterns around to see what kind of patterns you can make.

Can you make the FRAME pattern using squares or your cutouts?

Once you figure that one out, then figure out what the turtle would have to do to fill the FRAME pattern after it draws the first TOWER pattern?

_____

Back to Tower :S :T

When the turtle draws TOWER 15 4, it can't just turn around a draw the pattern again, can it? What would happen? Why not try it and see?

After the turtle gets to the top of the first pattern, it is going to have to move over a bit to draw the TOWER pattern coming down the screen. But how far?

You know that the pattern is :S steps wide. In TOWER 15 4, :S is 15, right? So let's write a procedure for the turtle to move at the top of the TOWER.

```
TO MOVE1
RT 90 FD _____ RT 90
END
```

When the turtle gets to the top of the TOWER, she'll turn right, move over, and then turn right again. What happens if we use the value of :S or 15? Does that work?

No. The turtle ends up drawing the pattern over the original drawing. When the turtle turns at the top, it starts drawing the SQUARES pattern by moving to the right. This means the turtle has to move twice as far, or :S * 2.

Try it. See what happens.

```
TO MOVE1 :S
RT 90 FD :S * 2 RT 90
END
```

Try this:

```
TOWER 15 4
MOVE1 15
```

It seems to work, doesn't it!



You're not out of the woods yet. Do you see that blank space at the bottom — to the right?

How are you going to fill that in? Also, what is the turtle going to have to do to draw the next TOWER?

How about this?

```
TO MOVE2 :S
RT 90 FD :S BK :S RT 90
END
```

The turtle turns right, fills in the gap, backs up, turns right again (that's 180 degrees), and is ready to start again.

You didn't know this was going to be this complicated, did you?

There's one more thing to do now. That's to write a procedure that will create the repeating tessellation.

We'll call it COVER.

```
TO COVER :S :T :X
IF :X = 0 [STOP]
TOWER :S :T
MOVE1 :S
TOWER :S :T
MOVE2 :S
```



```
COVER :S :T :X - 1
END
```

You already know what the :S and :T variables are. What about the :X?

That's easy enough. Just like the :T variable, :X tells COVER the number of times to repeat itself.

Musical Variables

In the last chapter, we talked about making music. Now that you've read about variables, how about some musical variables?

Do you want to turn your keyboard into musical keys? Here's one way to do it.

```
TO MUSIC
MAKE "KEY RC
IF :KEY = "C [SOUND [262 100]]
IF :KEY = "D [SOUND [294 100]]
IF :KEY = "E [SOUND [330 100]]
IF :KEY = "F [SOUND [349 100]]
IF :KEY = "G [SOUND [392 100]]
IF :KEY = "A [SOUND [440 100]]
IF :KEY = "B [SOUND [494 100]]
IF :KEY = "S [STOP]
MUSIC
END
```

There's another new command, RC. That's short for READCHAR. When Logo sees the READCHAR or RC command, it stops and waits for you to type a character. In this case, the letter you type becomes the variable :KEY.

If you type one of the keys — A, B, C, D, E, F, G — you hear a note. Just make sure you use a capital letter. Otherwise Logo just runs the MUSIC procedure again and again until you hit one of the sound keys and press Enter.

Rabbit Trail 18. Tangrams

The Tangram is an Oriental puzzle with seven shapes of different sizes.



The puzzle is to use these shapes to make lots of different things. Here's my pup tent.



Why not visit your local library or bookstore? You'll find there are a number of books on tangrams that will give you lots of ideas of what to do with your new puzzle pieces.

There's a PCX file on the Discovery book diskette called TANGRAM.PCX. There's a copy of the picture on the next page.



1. Print the picture and paste it to a piece of cardboard.
2. Carefully cut out the pieces.
3. Now you can play with the pieces to create interesting shapes: birds, ships, dragons, and other interesting designs.
4. Then draw them on the computer.

There's a procedure on the diskette that came with this book called TANGRAM.LGO. You can use that to create your

Tangram shapes. We talk about it in *The Great math Adventure* chapter.

Now, why not see what you can do with Tangrams?

Rabbit Trail 19. More on the Logo Puzzles

Remember the Logo Puzzles back in Chapter 2?

Get some straws or some sticks and make this puzzle on a table top. Now, take away just one straw or stick to make a picture that has only three squares.



You can solve this puzzle by picking up any of the sticks to see if three squares are left on the table. But we told you there was a procedure that would solve the puzzle for you. Let's see what it is.

First of all, let's write procedures to create the puzzle. Obviously, you'll need a SQUARE procedure.

```
TO SQUARE
REPEAT 4 [FD 100 RT 90]
END
```

Now you can write a PUZZLE procedure.

```
TO PUZZLE
CS HT
REPEAT 2 [SQUARE MOVE]
REPEAT 2 [SQUARE RT 180]
HOME
END

TO MOVE
RT 90 FD 100 LT 90
END
```

The next step is to solve the puzzle. But how could the computer do that? It doesn't think. It simply does what you tell it to do.

Since the computer does things much faster than you can, one way to have the computer help you solve the problem is to have it erase each line in the puzzle and then draw it again.

Sound confusing? Try this:

```
TO SOLVE
PUZZLE
REPEAT 2 [SQ MOVE]
REPEAT 2 [SQ RT 180]
HOME
END
```

```
TO SQ
REPEAT 4 [SIDE RT 90]
END

TO SIDE
PE FD 100 WAIT 100 PD (or PENPAINT)
PD BK 100 FD 100
END
```

Waiting

Do you remember when we mentioned "waiting" before? There are times that you want to slow down the computer so you can see what's going on, or when you just want it to wait a few seconds. That's where the WAIT command comes in.

WAIT *<time in 60ths of a second>*

There's another way to slow the computer down or to have it take a pause. Write your own WAIT command. Because WAIT is a primitive already, call your new procedure WAITS. Or maybe call it TIME or TIMER.

```
TO TIMER :T
IF :T = 0 [STOP]
TIME :T - 1
END
```

You can make this procedure as precise a timer as you need. You can make :T whatever you want. After all, it is a variable. You can also change :T - 1 to :T - 0.25 or whatever. It's another way to get Logo to do exactly what you want it to do.

"What if I just want to pause for a moment while running a procedure? Can I do that?"

Sure, you can. That's what the Pause button is for; the one over to the right in the Commander Box. Try it and see what happens.

And don't forget the SETTIMER command. This is described in the On-line Help file.

The Tacit Assumption

Here's another puzzle for you, the Tacit Assumption. Draw this pattern on paper or create it on the screen and print it

Your challenge is to draw four straight lines that pass through all nine of the star shapes. You can start anywhere you want. The catch is that once you put your pencil down, it cannot leave the paper until you are done.

.

Here's a hint. When most people try to solve this puzzle, they *assume* that they must stay within the limits of the square made by the nine stars. That's what we mean by the *Tacit* or unspoken *Assumption*.

The following procedure draws the nine points.

```
TO TACIT
CS HT PU
REPEAT 3 [LINE MOVE]
HOME
END

TO LINE
REPEAT 3 [ASTERISK FD 100] BK 300
END
TO MOVE
RT 90 FD 100 LT 90
END

TO ASTERISK
PD REPEAT 6 [FD 10 BK 10 RT 60] PU
END
```

Here's a procedure that will solve the puzzle. However, you have to figure out what number to use for the variable.

All you have to do is type SOLVE and add a guess. If your guess doesn't quite do it, type TACIT to draw the puzzle. Then guess again!

```
TO SOLVE :SIDE
FD :SIDE/SQRT 2 RT 135
FD :SIDE RT 135
FD :SIDE/SQRT 2 RT 90
RT 45 FD :SIDE/SQRT 2
END
```

That :SIDE / SQRT 2 may seem like something strange. But don't worry about it right now. It's just part of a math formula for drawing the long side of a right triangle.

You'll learn more about SQuare RooTs in *The Great Math Adventure* chapter.

The Non-Stop Puzzles

Remember the puzzles about the triangle patterns? Draw the patterns without retracing any line and without lifting the pen from the paper.



Here are the procedures. They give you some new things to think about.

```
TO TRIPUZZLE
CS CT
PR [HERE'S A PUZZLE FOR YOU!]
PR "
NONSTOP.TRI 100 WAIT 100 CT
PR [DRAW THIS PATTERN OF TRIANGLES
  WITHOUTLIFTING]
PR [YOUR PENCIL FROM THE PAPER AND
  WITHOUTRETRACING]
PR [ANY OF THE LINES.] WAIT 200 CT
```

```
PR [OR PRESS ANY KEY TO HAVE LOGO DO IT
  FOR YOU!]
IGNORE RC TRIANGLE 100 50
END
```

WOW! There are a bunch of things to look at here.

Here's another example of the PRINT command, or PR, for short. This time it's used to print instructions on what to do.

But what about that line, PR " ? What's that mean?

You can write it as PR " or PR []. What it says is to print nothing. And that's exactly what it does; it prints nothing, leaving you with a blank line.

And how about that line, IGNORE RC?

IGNORE is a command that does absolutely nothing! When you replace the :X variable with RC (or READCHAR) the procedure just stops and waits for you to press a key — to read the character or key that you press.

Now let's look at the procedures to draw the triangles. What's the difference between the two? They seem to do the same thing.

```
TO NONSTOP.TRI :D
CS HT LT 150 FD :D RT 120 FD :D * 3
RT 120 FD :D * 3 RT 120 FD :D * 2
RT 120 FD :D * 2 LT 120 FD :D
LT 120 FD :D * 2 LT 120 FD :D
LT 120 FD :D * 2 LT 120 FD :D
END
```

If you run the NONSTOP.TRI procedure on a fast computer, the drawing just sort of appears all at once. The turtle moves too fast for you to see how it drew the lines without retracing any of its steps.

Here's a procedure to solve that problem.

```
TO TRI :D :T
CS ST LT 150
FD :D RT 120 FD :D * 3 WAIT :T
RT 120 FD :D * 3 RT 120 FD :D * 2 WAIT :T
RT 120 FD :D RT 120 WAIT :T
REPEAT 3 [FD :D LT 120] LT 180 WAIT :T
REPEAT 3 [FD :D RT 120] WAIT :T
LT 60 FD :D LT 120 FD :D
END
```

The first difference you notice between NONSTOP.TRI and TRI is in the variables. TRI has an extra one, :T. It's used as an input to the WAIT command. In this way, you can add the amount of time you need to see how the drawing is made.

Pretty neat, huh?



Remember this puzzle?

Try this one, only draw it without crossing any line, without retracing any line, or lifting your pencil from the paper.

We added a procedure to solve this puzzle for you.

```
TO SOLVE
CS CT
```

```
PR [OK!  HOW ABOUT TRYING THE SAME ~
  THING WITH THIS HOUSE?  ONLY DON'T ~
  CROSS ANY LINES.]
HOUSE 100 WAIT 200 CT
PR [OR PRESS ANY KEY TO HAVE LOGO DO IT~
    FOR YOU!]
IGNORE RC
NONSTOP.HOUSE 100
END

TO HOUSE :D
CS FD :D LT 30
REPEAT 3 [FD :D LT 120]
LT 105 FD :D * .71 LT 90
FD :D * .71 RT 135 FD :D
RT 90 FD :D RT 135
FD :D * .71 RT 90
FD :D * .71
END
```

HOUSE :D draws the house. NONSTOP.HOUSE adds a time variable so you can see how it's drawn.

```
TO NONSTOP.HOUSE :D :T
CS FD :D LT 30
REPEAT 3 [FD :D LT 120] WAIT :T
LT 105 FD :D * .71 LT 90 WAIT :T
FD :D * .71 RT 135 FD :D WAIT :T
RT 90 FD :D RT 135 WAIT :T
FD :D * .71 RT 90 WAIT :T
FD :D * .71
END
```

Enough of this house business.

Adding Borders

Morf just loves to put borders around things, even the graphics window. Take a look!

Joe Power, a friend from California, taught Morf how to do that. It comes in real handy when you want to do a pretty card or announcement.

Here's the procedure.

```
TO BORDER
CS HT
PU SETXY -200 -100 PD
BRAID
END
```

You can change the coordinates where the procedures begins to make the border larger or smaller. You also have to change the last line of the BRAID procedure.

```
TO BRAID
MAKE "SQR2 1.4 ; SQRT 2
MAKE "HFSQ2 0.7 ; :SQR2 * 0.5
MAKE "S2 8.5 ; :SQR2 * 6
MAKE "H2 4.2 ; :HFSQ2 * 6
MAKE "S2H2 12.7 ; :S2 + :H2
PU FD 24 RT 45 FD 4.2 SETH 0 PD
REPEAT 2 [STRIP 20 CORNER STRIP 30 CORNER]
END
```

To change the size of the border, change the number of times that STRIP is repeated. Go ahead. Give it a try.

```
TO CORNER
LT 45 FD :H2 RT 45 FD 6
RT 45 FD :S2 RT 45 FD 18
RT 45 FD :S2H2 PU
RT 90 FD :H2 PD RT 90 FD :S2
LT 45 FD 18 LT 90 FD 6 PU
LT 45 FD :S2 PD LT 90 FD 17 PU
RT 90 FD :H2 PD RT 90 FD 17 PU
RT 45 FD 6 RT 90 FD 12 PD
RT 45 FD :H2 RT 45 FD 6
RT 45 FD :H2 PU RT 90 FD :H2 PD
RT 45 FD 6 PU BK 15 RT 90 FD 9 RT 90 PD
END

TO START
; Here's a simple procedure that puts a braided border
; around the edge of the screen.  Morf likes frames
; for his pictures.
; You can change the size of the border by changing the
; variable used by STRIP in the BRAID procedure.
```

```
BORDER
END

TO STRIP :N
REPEAT :N ~
[
LT 45 FD :H2 RT 45 FD 6 RT 45 FD :S2H2
PU RT 90 FD :H2 PD RT 90 FD :S2 LT 45 FD 6 PU
LT 45 FD :S2H2 PD LT 135 RT 45 FD :H2 LT 45
FD 6 LT 45 FD :S2H2 PU LT 90 FD :H2 PD LT 90
FD :S2 RT 45 FD 6 PU RT 135 FD :S2H2 RT 45
FD 6 PD
]
END
```

The Strip procedure is one long line. But look how it's written.

```
REPEAT :N ~
```

What's that symbol after :N?

It's a tilde. In MSW Logo, that means that the instruction list is continued on the next line. There you find a single

```
[
```

When you have long lines and lists inside other lists, they can get confusing — very difficult to read. When MSW Logo sees a single bracket like that, it knows to look on the next line for the rest of the list.

The rest of the line in STRIP is simply a long list of commands. But what if you had lists within lists. Here's a simple example.

```
TO HEX
REPEAT 6 ~
  [
    REPEAT 3 ~
          [
            FD 100 RT 120
          ]
    RT 60
  ]
END
```

This is the same as

```
TO HEX
REPEAT 6 [REPEAT 3 [FD 100 RT 120] RT 60]
END
```

When procedures begin to get long and complex, you need a system that allows you to read and understand what's going on. As you will see in coming chapters, this can come in real handy.

Check out the procedures in the MSW Logo Examples directory for some other examples of multi-line procedures.

If you think this is serious business, now it's time to get really serious about turtle geometry, starting with the next chapter.

But before you get there, here's a few more things variable things to explore.

Planting Another Garden

Early in this chapter, you had the chance to "plant another garden." Before you leave this chapter on variables, how about planting another garden by adding a twist to the Anyshape procedure. This also adds a twist to running procedures automatically and shows you something else about variables.

In the FLOWERS procedure, you run procedures from within another procedure. Take a look.

```
TO FLOWERS :REPEATS :LIST
REPEAT :REPEATS ~
   [RUN :LIST RT 360 / :REPEATS ]
END
```

RUN is a command that tells Logo to run a list of commands. Your remember what a list is, don't you?

The GARDEN procedure gives you a pretty good idea. Lists can contain words, commands, or other lists.

```
TO GARDEN
CS FLOWERS 5 [FD 50]
CS FLOWERS 5 [FD 60 POLY 50 5]
CS FLOWERS 5 [FD 50 LT 30]
CS FLOWERS 7 ~
      [FD 50 LT 60 FD 50 RT 120 FD 50 LT 60 FD 50]
CS FLOWERS 8 [POLY 100 5]
CS FLOWERS 8 [POLY 100 3]
CS FLOWERS 8 [POLY 100 4]
CS FLOWERS 8 [POLY 80 6]
CS FLOWERS 5 ~
          [FD 80 FLOWERS 8 [POLY 80 3] BK 80]
END

TO POLY :SIZE :REPEATS
REPEAT :REPEATS [FD :SIZE RT 360 / :REPEATS]
END
```

Last Minute Ideas

The GARDEN procedure is OK. But have you ever seen a black and white garden? Try adding some color to it.

GARDEN shows you a number of individual flower shapes. Maybe you want to change those shapes. Or maybe you want them to stay on the screen for a longer time. Add a WAIT command.

Remember the last FLOWER picture that is displayed?

```
FLOWERS 5 [FD 80 FLOWERS 8 [POLY 80 3] BK 80]
```

Why not add some variations of this to the GARDEN procedure so you have different groups of flowers in your garden. Here's one idea:

```
FLOWERS 12 [POLY 30 8]
```

Also, why not have your flower garden "grow" when it loads.

```
Make "startup [GARDEN]
```

Whatever you do, have fun with your new garden.
